
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of IMS Video Call test case 12.21 which is part of the TTCN-3 test suite in the ‘iwd-TTCN3-B2013-03_D14wk37’ ATS delivery.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
TC_12_21
Test Group:
IMS\12\ IMS_CC_CallControlTestcases.ttcn

ATS Version:
iwd-TTCN3-B2013-03_D14wk37
System Simulator used:
R&S CMW500 
UE used:
Qualcomm MSM 8974 Pro
Verification Status:
PASS


4. Corrections required for TC 12.21
Change 1  –  IPCAN / Dedicated bearer establishment
Change 1.1 – Introduction of new local function ‘fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_VideoCall’
	Function name
	fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_VideoCall ()

	Reason for change
	New function is needed for adding the two DRBs for audio and video according to  TS 36.508 4.5A.8 Generic Test Procedure for IMS MO video call establishment in E-UTRA, which specifies:
“One EPS bearer context #4 (QCI 1) and one EPS bearer context #3 (QCI 2) according table 6.6.2-1: Reference dedicated EPS bearer contexts.”

	Summary of change
	New function in coded in analogy to existing and verified function “fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_SpeechCall ()”

	TTCN module
	IPCAN_EUTRA.ttcn

	MCC160 Comment
	


Before change

After change

...

  // New function

  function fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_VideoCall(EUTRA_CellId_Type p_CellId

                                                                ) runs on EUTRA_PTC

  {

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    /* See TS 36.508 4.5A.8 Generic Test Procedure for IMS MO video call establishment in E-UTRA

      One EPS bearer context #4 (QCI 1) and one EPS bearer context #3 (QCI 2) according table 6.6.2-1: 

      Reference dedicated EPS bearer contexts.

     */

    fl_EUTRA_TxSpeechVideoEpsBearerCtxtReq(p_CellId,

                                        v_EPS_TI,

                                        cs_LinkedId(hex2bit(tsc_EpsDefaultBearerId)),

                                        tsc_EpsDedicatedBearerId, // Audio

                                        cds_EPS_QoS_Dedicated_4, 

                                        f_EUTRA_Get_508_TrafficFlowTemplate(4, tsc_EpsDedicatedBearerId),

                                        tsc_EpsDedicatedBearerId2, // Video

                                        cds_EPS_QoS_Dedicated_3, 

                                        f_EUTRA_Get_508_TrafficFlowTemplate(3, tsc_EpsDedicatedBearerId2)

                                        );  

    /* The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message for 

      each of previously sent ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST message

     */

    // FFS The messages may arrive in a single UL Information transfer

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(tsc_EpsDedicatedBearerId2))));

  }

  function fl_EUTRA_TxSpeechVideoEpsBearerCtxtReq(EUTRA_CellId_Type p_CellId,

                                               template (value) ProcedureTransactionIdentifier p_ProcTi,

                                               template (value) LinkedEpsBearerIdentity p_LinkedId, 

                                               EPS_BearerIdentity p_EPS_Id_Speech,

                                               template (value) EPS_QualityOfService p_QoS_Speech,

                                               template (value) TrafficFlowTemplate p_Tft_Speech,

                                               EPS_BearerIdentity p_EPS_Id_Video,

                                               template (value) EPS_QualityOfService p_QoS_Video,

                                               template (value) TrafficFlowTemplate p_Tft_Video

                                               )

    runs on EUTRA_PTC

  { 

    var template (omit) DRX_Config v_DRX_Config;

     var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var DRB_Identity v_DrbId_Speech := f_EUTRA_EpsBearerAssociatedDRB(p_EPS_Id_Speech);

    var DRB_Identity v_DrbId_Video := f_EUTRA_EpsBearerAssociatedDRB(p_EPS_Id_Video);

    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated;

    var template (omit)  NAS_MSG_RequestList_Type v_NAS_MsgList;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

   v_NAS_MsgList[0] := cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                           cs_508_ActDedicatedEPSBearerCxtReq(p_EPS_Id_Speech,

                                                 p_ProcTi,

                                                 p_LinkedId,

                                                 p_QoS_Speech,

                                                 p_Tft_Speech));  

   v_NAS_MsgList[1] := cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                           cs_508_ActDedicatedEPSBearerCxtReq(p_EPS_Id_Video,

                                                 p_ProcTi,

                                                 p_LinkedId,

                                                 p_QoS_Video,

                                                 p_Tft_Video));  

     v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_DEFAULT_UM (v_DrbId_Speech);

     v_DRB_ToAddModList[1] := cs_508_DRB_ToAddMod_DEFAULT_UM (v_DrbId_Video);

    v_RadioResourceConfigDedicated := cs_RadioResourceConfig_DRB_RECONFIG(

                                    {v_DRB_ToAddModList[0], v_DRB_ToAddModList[1]},

                                        cs_508_MAC_MainConfig_Explicit_RBC_DrxS);

    v_DRX_Config := f_GetUeDrxConfig(DRX_S);

    if (isvalue(v_DRX_Config)) {

        f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_Config(valueof(v_DRX_Config)));

    }

    SRB.send(cas_SRB1_RrcNasPduList_REQ(p_CellId,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_Common(v_RRC_TI,

                                                                           omit,

                                                                           omit,

                                                                           v_RadioResourceConfigDedicated,

                                                                           omit),

                                    v_NAS_MsgList));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(p_CellId,

                                    cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

  }
... 
Change 1.2 – Correction to function ‘f_EUTRA_IPCAN_MO_VideoCall’
	Function name
	f_EUTRA_IPCAN_MO_VideoCall ()

	Reason for change
	Activation of Dedicated EPS Bearer for Video Call is missing.

	Summary of change
	Call of newly defined local function (see change 1.1) “fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_VideoCall” is added which replaces the call to “fl_EUTRA_TxDedEpsBearers_Video” (see also change 1.3).

@ MCC160:
Even though not related to TC 12.21 but rather to TC 12.22, the newly defined local function should also be called in a similar way in the equivalent function for MT Video Calls (-->  f_EUTRA_IPCAN_MT_VideoCall).

	TTCN module
	IPCAN_EUTRA.ttcn

	MCC160 Comment
	


Before change

...
function f_EUTRA_IPCAN_MO_VideoCall(IMS_IPCAN_CO_ORD_PORT p_Port,

                                      EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC return TriggerResult_Type

  {

    var TriggerResult_Type v_Result := NORMAL;

    /* Table 4.5A.8.3-1 Step 2-8 */

    fl_EUTRA_IPCAN_ServiceRequestAndActivate_SRB2_DRB(p_CellId, cr_EstablishmentCause_MoData);

    v_Result := f_IMS_IPCAN_WaitForTriggerOrAbort(p_Port);  // wait until IMS PTC has sent "100 Trying" and "183 Session Progress" (steps 3 and 4 of C.21)

    if (v_Result == NORMAL) {                               // => procedure shall not be aborted

      fl_EUTRA_TxDedEpsBearers_Video(p_CellId);
    }

    return v_Result;

  }
... 
After change

...

function f_EUTRA_IPCAN_MO_VideoCall(IMS_IPCAN_CO_ORD_PORT p_Port,

                                      EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC return TriggerResult_Type

  {

    var TriggerResult_Type v_Result := NORMAL;

    /* Table 4.5A.8.3-1 Step 2-8 */

    fl_EUTRA_IPCAN_ServiceRequestAndActivate_SRB2_DRB(p_CellId, cr_EstablishmentCause_MoData);

    v_Result := f_IMS_IPCAN_WaitForTriggerOrAbort(p_Port);  // wait until IMS PTC has sent "100 Trying" and "183 Session Progress" (steps 3 and 4 of C.21)

    if (v_Result == NORMAL) {                               // => procedure shall not be aborted

      fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_VideoCall (p_CellId);
    }

    return v_Result;

  }
... 
Change 1.3 – Removal of ‘fl_EUTRA_TxDedEpsBearers_Video’
	Function name
	fl_EUTRA_TxDedEpsBearers_Video ()

	Reason for change
	“f_EUTRA_RRCConnectionReconfiguration_Common” can not be used for adding the two DRBs needed for video and speech, as the function “f_EUTRA_RRCConnectionReconfiguration_Common” adds also the default AM DRB which is already established.

Note that DRBs and EPS Bearers have to match TS 36.508 4.5A.8.

	Summary of change
	Function removed as it was replaced by newly introduced function in change 1.1.

	TTCN module
	IPCAN_EUTRA.ttcn

	MCC160 Comment
	


Before change

...
  /*

   * @desc      This function configures the dediated bearers required for Video

   * @param     p_CellId

   * @status    APPROVED (IMS)

   */

  function fl_EUTRA_TxDedEpsBearers_Video(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC

  {

    var template (value) EPS_QoSList_Type v_Qos_List := {cds_EPS_QoS_Dedicated_3, cds_EPS_QoS_Dedicated_4};

    var template (value) Tft_List_Type v_Tft_List := {f_EUTRA_Get_508_TrafficFlowTemplate(3, tsc_EpsDedicatedBearerId), f_EUTRA_Get_508_TrafficFlowTemplate(4, int2hex(hex2int(tsc_EpsDedicatedBearerId)+1, 1))};

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    // send RRCConnectionReconfiguration

    f_EUTRA_RRCConnectionReconfiguration_Common(p_CellId, v_RRC_TI, 0, 2, -, -, -, -, v_Qos_List, v_Tft_List);

    // receive RRCConnectionReconfigurationComplete

    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(p_CellId, v_RRC_TI, 0, 2);

  }
... 
After change

Change 2  –  Receive templates used in step 2 of Annex C.25

The following receive templates need to be corrected. They are used for matching the video SDP record contents in the initial INVITE  message send by the UE step 2 in Annex C.25.
Change 2.1 – Correction to template ‘cr_Fmtp_VideoParamsWithPacketizationMode’
	Template name
	cr_Fmtp_VideoParamsWithPacketizationMode ()

	Reason for change
	· Packetization mode 1 was incorrectly used in the expected template. 
· Profile-level-id 42e00c  was expected.

	Summary of change
	Step 2 in Annex C.25 does not specify the expected packetization mode nor the profile-level-id.

	TTCN module
	IMS_SDP_Templates.ttcn

	MCC160 Comment
	


Before change

...

template (present) SemicolonParam_List cr_Fmtp_VideoParamsWithPacketizationMode := {cr_Video_Fmtp_PacketizationMode, cr_Video_Fmtp_ProfileLevelId(tsc_IMS_Video_Fmtp_ProfileLevelId_Def), cr_Video_Fmtp_SpropParameterSets, *};
...

After change

...

template (present) SemicolonParam_List cr_Fmtp_VideoParamsWithPacketizationMode := {cr_Video_Fmtp_PacketizationMode, cr_Video_Fmtp_ProfileLevelId(?), cr_Video_Fmtp_SpropParameterSets, *};
... 
Change 2.2 – Correction to template ‘cr_Video_Fmtp_SpropParameterSets’
	Template name
	cr_Video_Fmtp_SpropParameterSets ()

	Reason for change
	The expected value of the the parameter sprop-parameter-sets was set to J0LgDJWgUH6Af1A=,KM46gA==.  Following a discussion with the author of the Prose Specification of this TC, this is incorrect since this parameter is optional. 

(see associated Prose CR “draft_R5-150xxx_TS-34229-1_Annex-C-25.zip”).

	Summary of change
	In the fmtp attribute for video SDP records, any value in the parameter sprop-parameter-sets is accepted.

	TTCN module
	IMS_SDP_Templates.ttcn

	MCC160 Comment
	


Before change

...

template (present) GenericParam cr_Video_Fmtp_SpropParameterSets := f_GenericParam_AsTX(cs_Video_Fmtp_SpropParameterSets);
...

After change

...

template (present) GenericParam cr_Video_Fmtp_SpropParameterSets := 

cr_GenericParam("sprop-parameter-sets", ?);
... 
Change 2.3 – Correction to template ‘cr_SDP_Media_Video’
	Template name
	cr_SDP_Media_Video ()

	Reason for change
	The template used incorrect transport “RTP/AVP” in the video media line.

	Summary of change
	Transport format corrected to “RTP/AVPF”, see Annex C.25 step 2.

	TTCN module
	IMS_SDP_Templates.ttcn

	MCC160 Comment
	


Before change

...

template (present) SDP_media_field cr_SDP_Media_Video(template (present) charstring p_Fmt := cr_SDP_AnyMediaFormatDescr) :=

    cr_SDP_Media(c_video, -, -, cr_SDP_FmtList_AtLeastOne(p_Fmt));
...

After change

...

template (present) SDP_media_field cr_SDP_Media_Video(template (present) charstring p_Fmt := cr_SDP_AnyMediaFormatDescr) :=

    cr_SDP_Media(c_video, -, c_rtpAvpf, cr_SDP_FmtList_AtLeastOne(p_Fmt));
... 
Change 3  –  Send templates used in step 4 of Annex C.25

The following send templates need to be corrected. They are used for sending the video SDP record contents in the initial 183 Sesion In Progress message at step 4 in Annex C.25.
Change 3.1 – Correction to function ‘f_IMS_BuildSDP_AnnexC25_Step4’
	Function name
	f_IMS_BuildSDP_AnnexC25_Step4 ()

	Reason for change
	· The fmtp parameters were not set according to step 4 in Annex C.25.

· Video attributes to be sent in the SDP video offer were not complete.

	Summary of change
	· The fmtp parameters are copied from the values sent by the UE in the initial Invite message.

· SDP records completed according to Annex C.25 step 4.

	TTCN module 
	IMS_CC_CallControlTestcases.ttcn

	MCC160 Comment
	


Before change

...

  function f_IMS_BuildSDP_AnnexC25_Step4(SDP_media_desc p_SDP_Media_AudioRx,

                                         SDP_media_desc p_SDP_Media_VideoRx,

                                         boolean p_TcapPcfg) runs on IMS_PTC return template (value) SDP_Message

  { /* Build SDP record to be sent to the UE at step 4 according to 34.229-1 Annex C.25 */

    /* @sic R5-144402: p_IPAddrOrigin removed sic@ */

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var charstring v_IPAddrConnection := f_IMS_PTC_RemoteAddress_GetAddrStr();

    var charstring v_IPAddrOrigin := v_IPAddrConnection;

    var template (value) SDP_media_desc v_SDP_Media_Desc_AudioTx;

    var template (value) SDP_media_desc v_SDP_Media_Desc_VideoTx;

    var template (value) SDP_media_desc_list v_SDP_Media_Desc_List;

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Audio := p_SDP_Media_AudioRx.bandwidth;

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Video := p_SDP_Media_VideoRx.bandwidth;

    var charstring v_FmtAudio := f_SDP_MediaDescr_GetAMR_8000_1(p_SDP_Media_AudioRx);

    var charstring v_FmtVideo := f_SDP_MediaDescr_GetH264_90000(p_SDP_Media_VideoRx);

    var template (value) SDP_attribute_list v_SDP_Attribute_List_Audio := {

      // media attributes:

      cs_SDP_Attribute_rtpmap(v_FmtAudio, cs_RTPMAP_AMR_8000),

      cs_SDP_Attribute_fmtp(v_FmtAudio, cs_AMR_Fmtp_DefaultParameters),
      cs_SDP_Attribute_ptime,

      cs_SDP_Attribute_maxptime,

      // precondition attributes:

      cs_SDP_Attribute_curr_qos(c_local),

      cs_SDP_Attribute_curr_qos(c_remote),

      cs_SDP_Attribute_des_qos(c_mandatory, c_local),

      cs_SDP_Attribute_des_qos(c_mandatory, c_remote),

      cs_SDP_Attribute_conf_qos

    };

    var template (value) SDP_attribute_list v_SDP_Attribute_List_Video := {

      // media attributes:

      cs_SDP_Attribute_rtpmap(v_FmtVideo, cs_RTPMAP_Codec("H264", "")),        // !!!! codec need to be specified in the prose !!!!

      cs_SDP_Attribute_fmtp(v_FmtVideo {cs_GenericParam("", "")})),   // !!!! parameters need to be specified in the prose !!!!
      cs_SDP_Attribute_ptime,

      cs_SDP_Attribute_maxptime,

      // precondition attributes:

      cs_SDP_Attribute_curr_qos(c_local),

      cs_SDP_Attribute_curr_qos(c_remote),

      cs_SDP_Attribute_des_qos(c_mandatory, c_local),

      cs_SDP_Attribute_des_qos(c_mandatory, c_remote),

      cs_SDP_Attribute_conf_qos

    };

    v_SDP_Media_Desc_AudioTx := cs_SDP_Media_Desc(cs_SDP_Media_Audio(v_FmtAudio),

                                                  v_Bandwidth_List_Audio,

                                                  v_SDP_Attribute_List_Audio);

    v_SDP_Media_Desc_VideoTx := cs_SDP_Media_Desc(cs_SDP_Media_Video(v_FmtVideo),

                                                  v_Bandwidth_List_Video,

                                                  v_SDP_Attribute_List_Video);

    v_SDP_Media_Desc_VideoTx.attributes := f_SDP_Attributes_Add_TX(v_SDP_Media_Desc_VideoTx.attributes, cs_SDP_Attribute_acfg(1, "t=1"), p_TcapPcfg);

    v_SDP_Media_Desc_List := { v_SDP_Media_Desc_AudioTx, v_SDP_Media_Desc_VideoTx };
    return cs_SDP_Message_Common(cs_SDP_Origin(-, -, v_AddrType, v_IPAddrOrigin), cs_SDP_Connection(v_AddrType, v_IPAddrConnection), -, cs_SDP_Bandwidth_List_AS(30), -, v_SDP_Media_Desc_List); /* @sic R5-144402: v_Sessionname removed sic@ */

  }
...

After change

...

  function f_IMS_BuildSDP_AnnexC25_Step4(SDP_media_desc p_SDP_Media_AudioRx,

                                         SDP_media_desc p_SDP_Media_VideoRx,

                                         boolean p_TcapPcfg) runs on IMS_PTC return template (value) SDP_Message

  { /* Build SDP record to be sent to the UE at step 4 according to 34.229-1 Annex C.25 */

    /* @sic R5-144402: p_IPAddrOrigin removed sic@ */

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var charstring v_IPAddrConnection := f_IMS_PTC_RemoteAddress_GetAddrStr();

    var charstring v_IPAddrOrigin := v_IPAddrConnection;

    var template (value) SDP_media_desc v_SDP_Media_Desc_AudioTx;

    var template (value) SDP_media_desc v_SDP_Media_Desc_VideoTx;

    var template (value) SDP_media_desc_list v_SDP_Media_Desc_List;

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Audio := p_SDP_Media_AudioRx.bandwidth;

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Video := p_SDP_Media_VideoRx.bandwidth;

    var charstring v_FmtAudio := f_SDP_MediaDescr_GetAMR_8000_1(p_SDP_Media_AudioRx);

    var charstring v_FmtVideo := f_SDP_MediaDescr_GetH264_90000(p_SDP_Media_VideoRx);

  var template SDP_attribute  v_Attribute_Video_fmtp := f_SDP_AttributeList_GetAttribute(p_SDP_Media_VideoRx.attributes, cr_SDP_Attribute_fmtp(v_FmtVideo, cr_Video_Fmtp_DefaultParameters));
    var template (value) SDP_attribute_list v_SDP_Attribute_List_Audio := {

      // media attributes:

      cs_SDP_Attribute_rtpmap(v_FmtAudio, cs_RTPMAP_AMR_8000),

     v_Attribute_Video_fmtp, // Copy received fmtp parameters 

      cs_SDP_Attribute_ptime,

      cs_SDP_Attribute_maxptime,

      // precondition attributes:

      cs_SDP_Attribute_curr_qos(c_local),

      cs_SDP_Attribute_curr_qos(c_remote),

      cs_SDP_Attribute_des_qos(c_mandatory, c_local),

      cs_SDP_Attribute_des_qos(c_mandatory, c_remote),

      cs_SDP_Attribute_conf_qos

    };

    var template (value) SDP_attribute_list v_SDP_Attribute_List_Video := {

      // media attributes:

      cs_SDP_Attribute_rtpmap(v_FmtVideo, cs_RTPMAP_Codec("H264", "90000")),    

      cs_SDP_Attribute_fmtp(v_FmtVideo, cs_Fmtp_ParamList(cs_Fmtp_VideoParamsWithPacketizationMode)),   

      cs_SDP_Attribute_ptime,

      cs_SDP_Attribute_maxptime,

      // precondition attributes:

      cs_SDP_Attribute_curr_qos(c_local),

      cs_SDP_Attribute_curr_qos(c_remote),

      cs_SDP_Attribute_des_qos(c_mandatory, c_local),

      cs_SDP_Attribute_des_qos(c_mandatory, c_remote),

      cs_SDP_Attribute_conf_qos

    };

    v_SDP_Media_Desc_AudioTx := cs_SDP_Media_Desc(cs_SDP_Media_Audio(v_FmtAudio),

                                                  v_Bandwidth_List_Audio,

                                                  v_SDP_Attribute_List_Audio);

    v_SDP_Media_Desc_VideoTx := cs_SDP_Media_Desc(cs_SDP_Media_Video(v_FmtVideo),

                                                  v_Bandwidth_List_Video,

                                                  v_SDP_Attribute_List_Video);

    v_SDP_Media_Desc_VideoTx.attributes := f_SDP_Attributes_Add_TX(v_SDP_Media_Desc_VideoTx.attributes, cs_SDP_Attribute_acfg(1, "t=1"), p_TcapPcfg);

    v_SDP_Media_Desc_List := { v_SDP_Media_Desc_AudioTx, v_SDP_Media_Desc_VideoTx };

    return cs_SDP_Message_Common(cs_SDP_Origin(-, -, v_AddrType, v_IPAddrOrigin), cs_SDP_Connection(v_AddrType, v_IPAddrConnection), -, cs_SDP_Bandwidth_List_AS(30), -, v_SDP_Media_Desc_List); /* @sic R5-144402: v_Sessionname removed sic@ */

  }
... 
Change 4  –  Receive templates used in step 7 of Annex C.25

The following receive templates need to be corrected. They are used for matching the video SDP record contents in in the initial UPDATE message send by the UE in step 7 of Annex C.25.
Change 4.1 – Correction to function ‘f_IMS_BuildSDP_AnnexC25_Step5or7’
	Function name
	f_IMS_BuildSDP_AnnexC25_Step5or7 ()

	Reason for change
	Attribute a=des:qos in the received media description for video does not correspond to the prose in Annex C.25 step 7. The UE may send ‘strength’ as “be mandatory OR opitional”.

	Summary of change
	Expected SDP record corrected, the ‘strength’ may be mandatory OR opional.

	TTCN module

	IMS_CC_CallControlTestcases.ttcn

	MCC160 Comment
	


Before change

...

  function f_IMS_BuildSDP_AnnexC25_Step5or7(charstring p_FmtAudio,

                                            charstring p_FmtVideo) runs on IMS_PTC return template (present) SDP_Message

  { /* to receive 2nd SDP answer sent by the UE in C.25 step 5/7 */

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrAudio := ?;

    var template (present) charstring v_IPAddrVideo := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc_Audio;

    var template (present) SDP_media_desc v_SDP_Media_Desc_Video;

    var template (present) SDP_media_desc_list v_SDP_Media_Desc_List;

    var template (present) SDP_attribute_list v_SDP_Attribute_List_Audio := {

      // media attributes:

      cr_SDP_Attribute_rtpmap(p_FmtAudio, cr_RTPMAP_AMR_8000),     // @sic R5s130681 change 2 sic@

      cr_SDP_Attribute_fmtp(p_FmtAudio),

      cr_SDP_Attribute_sendrecv,

      // precondition attributes:

      cr_SDP_Attribute_curr_qos(c_local, c_sendrecv),

      cr_SDP_Attribute_curr_qos(c_remote),

      cr_SDP_Attribute_des_qos(c_mandatory, c_local),

      cr_SDP_Attribute_des_qos((c_optional, c_mandatory), c_remote),  /* @sic R5-142951: (c_optional, c_mandatory) instead of c_optional sic@ */

      *

    };

    var template (present) SDP_attribute_list v_SDP_Attribute_List_Video := {

      // media attributes:

      cr_SDP_Attribute_rtpmap(p_FmtVideo, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(p_FmtVideo, cr_Video_Fmtp_DefaultParameters),

      // precondition attributes:

      cr_SDP_Attribute_curr_qos(c_local, c_sendrecv),

      cr_SDP_Attribute_curr_qos(c_remote),

      cr_SDP_Attribute_des_qos(c_mandatory, c_local),

      cr_SDP_Attribute_des_qos(c_optional, c_remote), 

      *

    };

    v_SDP_Media_Desc_Audio := cr_SDP_Media_Desc(cr_SDP_Media_Audio(p_FmtAudio),

                                                cr_SDP_Connection(v_AddrType, v_IPAddrAudio),

                                                cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                                v_SDP_Attribute_List_Audio);

    v_SDP_Media_Desc_Video := cr_SDP_Media_Desc(cr_SDP_Media_Video(p_FmtVideo),

                                                cr_SDP_Connection(v_AddrType, v_IPAddrVideo),

                                                cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                                v_SDP_Attribute_List_Video);

    v_SDP_Media_Desc_List := { v_SDP_Media_Desc_Audio, v_SDP_Media_Desc_Video };

    return cr_SDP_Message_Common(v_AddrType, v_IPAddrOrigin, ?, -, -, cr_SDP_Time_Any, v_SDP_Media_Desc_List);

  }
...

After change

...

  function f_IMS_BuildSDP_AnnexC25_Step5or7(charstring p_FmtAudio,

                                            charstring p_FmtVideo) runs on IMS_PTC return template (present) SDP_Message

  { /* to receive 2nd SDP answer sent by the UE in C.25 step 5/7 */

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrAudio := ?;

    var template (present) charstring v_IPAddrVideo := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc_Audio;

    var template (present) SDP_media_desc v_SDP_Media_Desc_Video;

    var template (present) SDP_media_desc_list v_SDP_Media_Desc_List;

    var template (present) SDP_attribute_list v_SDP_Attribute_List_Audio := {

      // media attributes:

      cr_SDP_Attribute_rtpmap(p_FmtAudio, cr_RTPMAP_AMR_8000),     // @sic R5s130681 change 2 sic@

      cr_SDP_Attribute_fmtp(p_FmtAudio),

      cr_SDP_Attribute_sendrecv,

      // precondition attributes:

      cr_SDP_Attribute_curr_qos(c_local, c_sendrecv),

      cr_SDP_Attribute_curr_qos(c_remote),

      cr_SDP_Attribute_des_qos(c_mandatory, c_local),

      cr_SDP_Attribute_des_qos((c_optional, c_mandatory), c_remote),  /* @sic R5-142951: (c_optional, c_mandatory) instead of c_optional sic@ */

      *

    };

    var template (present) SDP_attribute_list v_SDP_Attribute_List_Video := {

      // media attributes:

      cr_SDP_Attribute_rtpmap(p_FmtVideo, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(p_FmtVideo, cr_Video_Fmtp_DefaultParameters),

      // precondition attributes:

      cr_SDP_Attribute_curr_qos(c_local, c_sendrecv),

      cr_SDP_Attribute_curr_qos(c_remote),

      cr_SDP_Attribute_des_qos(c_mandatory, c_local),

      cr_SDP_Attribute_des_qos((c_optional, c_mandatory), c_remote), 
      *

    };

    v_SDP_Media_Desc_Audio := cr_SDP_Media_Desc(cr_SDP_Media_Audio(p_FmtAudio),

                                                cr_SDP_Connection(v_AddrType, v_IPAddrAudio),

                                                cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                                v_SDP_Attribute_List_Audio);

    v_SDP_Media_Desc_Video := cr_SDP_Media_Desc(cr_SDP_Media_Video(p_FmtVideo),

                                                cr_SDP_Connection(v_AddrType, v_IPAddrVideo),

                                                cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                                v_SDP_Attribute_List_Video);

    v_SDP_Media_Desc_List := { v_SDP_Media_Desc_Audio, v_SDP_Media_Desc_Video };

    return cr_SDP_Message_Common(v_AddrType, v_IPAddrOrigin, ?, -, -, cr_SDP_Time_Any, v_SDP_Media_Desc_List);

  }
... 
5. Execution Log Files

5.1 Qualcomm MSM 8974 Pro
The Qualcomm MSM 8974 Pro UE passed this test case on R&S CMW500 in. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
Rohde-Schwarz\tc_IMS_12_21_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

	R5s141222:    Supporting information for agreement of IMS TC 12.21.

